ISSN 2076-8184. Інформаційні технології і засоби навчання. 2009. №6 (14).
Режим доступу до журналу: http://www.ime.edu-ua.net/em.html

УДК 517.53:004.021:378.147.227:372.851

Парфьонова Наталія Дмитрівна, кандидат фізико-математичних наук, доцент кафедри вищої математики фізичного факультету Харківського національного університету імені В.Н. Каразіна

Клочко Тамара Володимирівна, старший викладач кафедри вищої математики фізичного факультету Харківського національного університету імені В.Н. Каразіна

ОБЧИСЛЕННЯ ІНТЕГРАЛІВ ВІД ФУНКЦІЙ КОМПЛЕКСНОЇ ЗМІННОЇ ЗА ДОПОМОГОЮ ЗАЛИШКІВ В СИСТЕМАХ КОМП'ЮТЕРНОЇ МАТЕМАТИКИ

Анотація

В статті наведено методику викладання мабуть однієї з найважливіших тем курсу комплексного аналізу з використанням системи Maple; демонструється, як використання систем комп'ютерної математики полегшує викладачеві оцінювання роботи студентів та при організації їх самостійної роботи.

Ключові слова: Maple, математичного моделювання, інформаційні технології в навчанні.

Вступ. В останні роки отримав широкий розвиток новий науковий напрям – комп'ютерна математика. При цьому програмування залишається і надалі доволі складним і мало зрозумілим для студентів фізичних та інженерних спеціальностей. Більш простішими та доступними для них є системи комп'ютерної математики (надалі СКМ).

Зараз СКМ широко використовують для розв'язання наукових, інженерних, навчальних задач, наочної візуалізації даних і результатів обчислень та як зручні й повні довідники з математичних обчислень. Завдяки потужній графіці, засобам візуального програмування й використання техніки мультимедіа їх роль далеко виходить за межі тільки математичних розрахунків. Вони використовуються в освіті як потужні інструментальні засоби для підготовки електронних уроків, курсів лекцій та електронних книг з динамічними прикладами, які студенти самі можуть змінювати та виконувати навчальні дослідження. Використання СКМ дає змогу значною мірою підсилити інтелектуальну діяльність, дозволяє автоматизувати виконання не тільки чисельних, а й аналітичних (символьних) обчислень та графічних побудов.

Актуальність теми. СКМ активно використовуються у навчальному процесі у всьому світі. Виділяють сім основних класів СКМ: системи для чисельних обчислень, табличні процесори, матричні системи, системи для статистичних, для спеціальних обчислень, системи для аналітичних обчислень (комп'ютерної алгебри), універсальні системи. За останні кілька десятків років розроблено низку математичних пакетів як спеціалізованих (Eureca, MacMath, StatGraph, Reduse, MacSyma, SketchPad, Cabrs і ін.), так і універсальних (Derive, MathCad, MathLab, Maple, Mathematica, MuPad) [1-5] зі зручним інтерфейсом, в яких реалізовано значну кількість стандартних та спеціальних математичних операцій та функцій, потужні графічні засоби дво- та тривимірної графіки, власні мови програмування, засоби підготовки математичних текстів для друку, експортування даних в інші програмні продукти та імпортування з них даних для опрацювання. Все це забезпечує широкі можливості для ефективної роботи з пакетами фахівців різних профілів.

Підготовка майбутніх фізиків та інженерів до використання СКМ як в процесі навчання, так і в подальшій професійній діяльності набуває особливого значення. Тому розробка методик навчання математичних дисциплін з використанням СКМ, створення на їх основі інформаційних навчальних середовищ є досить актуальною проблемою.

Наявність різноманітних СКМ аж ніяк не означає, що успішно можна розв'язувати математичні задачі без відповідної теоретичної математичної підготовки, наявності вмінь розв'язувати задачі. Отже, СКМ є потужним засобом комп'ютерної підтримки діяльності науковців, учнів, студентів, педагогів, інженерів, але ефективність і методична цінність такого засобу залежить від вмінь застосовувати його. Тому проблема розробки методик навчання математичних дисциплін з використанням СКМ, гармонійне поєднання традиційних методичних систем навчання з сучасними комп'ютерними технологіями залишається актуальною.

Постановка проблеми. Знанням можна навчитися тільки в процесі їх використання в діяльності, тільки оперуючи ними. Придбати знання означає виконати з їх допомогою яку-небудь роботу, здійснити яку-небудь діяльність. Вони засвоюються для того, щоб з їх допомогою виконувати дії, здійснювати діяльність, а не для того, щоб вони просто запам'ятовувалися. Тому треба організувати навчальний процес так, щоб придбані знання стали засобом отримання нових знань.

На вивчення теорії функцій комплексної змінної в університетах на нематематичних спеціальностях, зазвичай, відводиться мало часу. Зокрема студенти зі слабою попередньою підготовкою не встигають засвоїти, мабуть, одну з найважливіших тем: «Обчислення інтегралів від функції комплексної змінної за допомогою залишків». Студентам завжди нелегко визначити особливі точки функції, намалювати контур, вибрати серед них ті, що потрапили всередину контуру, бо це потребує знань та вмінь зі шкільного курсу математики (треба розв'язати рівняння, іноді доволі складне), аналітичної геометрії (треба впізнати намалювати контур інтегрування). Але найскладніше, виявляється, визначити тип особливості в кожній точці. Цим зумовлена неможливість обчислення залишку, а надалі, інтеграла.

Наведемо опис методики викладання цієї теми для студентів фізичних та інженерних спеціальностей із застосуванням пакета Maple. Maple допоможе знайти особливі точки, намалювати контур, обчислити залишки та інтеграл. Зазначимо, що визначати тип особливості в більшості випадків не треба. Такий підхід дозволяє продемонструвати можливості математичного пакета й при цьому закріпити методи рішення завдань класичного математичного курсу.

Метою даної статті є показати ті програмні засоби, які можуть бути використані на практичних заняттях з теорії функцій комплексної змінної, які формують у студентів прийоми навчально-пізнавальної діяльності та сприяють розвитку творчого мислення.

Також хотілось би продемонструвати, як використання СКМ полегшує викладачеві оцінювання роботи студентів та організацію їх самостійної роботи.

Аналіз останніх досліджень і публікацій. Кількість публікацій, що стосуються опису та використанню пакету Maple, в світі дуже велика, але прямих аналогів ми знайти не змогли. При цьому в статті ми намагалися продемонструвати не лише можливості даного математичного пакету, бо для знайомства з ним існують досить гарні і дуже великі посібники, наприклад [1-5], а ті можливості, які дозволяють підвищити наочність учбового матеріалу та уникнути чорнових трудомістких розрахунків.

Обчислення інтегралів за допомогою залишків

Епізодичне необґрунтоване використання будь-якого математичного пакета не дає бажаних наслідків. При доборі СКМ слід враховувати також особливості задачі, що розв'язується. У даному випадку було обрано СКМ Maple, бо він є одним із світових лідерів серед пакетів комп'ютерної алгебри (Derive (Нова Зеландія); Mathematica (США); Maple (Канада); MathCAD (США)) і буде корисний для подальшого використання студентами при виконанні курсових та дипломних робіт і майбутньої професійної діяльності.

Нагадаймо, що для того, щоб аналітично обчислити контурний інтеграл від функції комплексної змінної необхідно:

1) знайти всі особливі точки функції;
2) вибрати серед них ті, що попали всередину контуру інтегрування;
3) визначити тип особливості в кожній точці;
4) в залежності від типу обчислити лишок (для кожного типу особливості – своя формула для обчислення залишку, див., наприклад [7, заняття 7], [6, пункти 1.7, 1.8]);
5) знайти значення інтеграла за теоремою Коші про залишки ([7, с. 10], [6, с. 63]).

Хоча, в деяких випадках, простіше отримати результат, використовуючи особливі точки, що не потрапили всередину контуру інтегрування (див. [4, приклад 19]). Користуючись системою Maple, будемо знаходити особливості та їх кількість за допомогою команд `singular` і `nops`, лишок – `residue`. У такий спосіб студенту не треба з'ясовувати тип особливості та знати формули, які їм відповідають для обчислення залишку. При цьому в істотно особливій точці все ж таки прийдеться зробити інакше (див. [7, приклад 20]).

На нашу думку треба показати студентам у будь-якій доступній формі: або на лекції (якщо вони передбачені програмою) або у роздрукованому вигляді на практичному занятті, надаючи кожному індивідуальні задачі, наприклад такі.

Обчислити за допомогою лишків наступні контурні інтеграли:

1) $\oint\limits_{|z-1-0.5i|=1.5} \dfrac{dz}{z^3(z^2+4)(z^4-1)}$; 2) $\oint\limits_{|z-1-i|=3} \dfrac{dz}{z^3(z^2+4)(z^4-1)}$;

3) $\oint\limits_{|z-1-0.5i|=2} \dfrac{(z+1)^{12}\,dz}{z^3(z^2+4)(z^4-1)}$; 4) $\oint\limits_{|z|=\pi} \cos\left(\cos\dfrac{1}{z}\right)dz$.

Задачі 1, 2, 4 оцінюються в 1 бал за кожну. Задача 3 оцінюється в 2 бали. Усього 3*1+2=5. Так до кінця заняття кожен студент має змогу оцінити себе сам.

Перші три задачі ззовні схожі, але різні по суті. Так, задачі 1 й 2 повинні бути виконані по теоремі Коші про залишки: перша – із використанням особливих точок підінтегральної функції, що потрапили усередину контуру інтегрування, друга – тих, що не потрапили всередину контуру інтегрування. Задача 3 відрізняється від попередніх тим, що в нескінченно віддаленій точці в підінтегральній функції полюс, а не усувна особливість. Викладач повинен звернути увагу студентів, які виконують цю задачу, що Maple-функція `singular` «налаштована» на дійснозначні функції, тому серед особливостей будуть плюс і мінус нескінченності.

У зв'язку з тим, що задача 3 найбільш трудомістка, варто на початку заняття порекомендувати студентам виконувати її після інших у час, що залишився. Всім бажаючим на наступному занятті можна продемонструвати рішення задачі, аналогічної до третій.

Далі наведемо приклади з повними текстами Maple-програм із методичними зауваженнями до кожної. Студент повинен вивести умови завдання та його розв'язання в «найгарнішому» вигляді й зробити перевірку отриманого розв'язку. Це привчає до охайності в роботі та позбавляє сліпої віри комп'ютерові.

Приклад 1. *Обчислити інтеграл* $\oint\limits_{|z+2i|=3} \dfrac{dz}{z^3(z^2+4)^2}$.

Введемо дані задачі, позначивши через `zc` – центр кола інтегрування, `R` – його радіус, **C** – коло інтегрування:

```
> with(plots): with(plottools):
> f:=z->1/((z^3)*(z^2+4)^2):'f(z)'=f(z);  zc:=-2*I: R:=3:C:=abs(z-zc)=R:
```

$$f(z) = \frac{1}{z^3(z^2+4)^2}$$

Згадаємо теорему Коші про залишки (підсумовування йде по всіх ізольованих особливих точках, які потрапили всередину контура інтегрування). Треба зауважити, що в пакеті прийнято позначати границі інтегрування дещо праворуч від інтегралу, а не чітко над і під ним.

```
>`Теорема Коші про залишки:`;
> Int('f(z)',z='C'..``)=2*Pi*I*Sum(res[z=z[k]]('f(z)'),k);
> Int(f(z),z=C..``)=2*Pi*I*Sum(res[z=z[k]](f(z)),k);
```

Теорема Коші про залишки:

$$\int_C f(z)dz = 2I\pi\left(\sum_k res_{z=z_k} f(z)\right)$$

$$\int_{|z+2i|=3} \frac{1}{z^3(z^2+4)^2}\,dz = 2I\pi\left(\sum_k \underset{z=z_k}{res}\left(\frac{1}{z^3(z^2+4)^2}\right)\right)$$

Знайдемо особливі точки даної функції за допомогою функції `singular` та їх кількість – функцією `nops`. Але функція `nops` вміє знаходити кількість елементів множини, або списку, в данному випадку – зручніше список.

```
> zzz:=[singular(f(z))]: n:=nops(zzz):
> `Особливі точки функції f(z):`;
> for j from 1 to n do
>    z[j]:=op(zzz[j][1])[2]: end do: print(seq(z||i,i=1..n));
```

Особливі точки функції f(z):

$$z_1 := 0 \quad z_2 := 2I \quad z_3 := -2I$$

Малюємо контур інтегрування та особливі точки.

```
> for j from 1 to n do
> pp||j:=textplot([(Re(z[j]))+0.4,Im(z[j])-0.1,"z"||j], align = BELOW):
> pp||(n+j):=circle([Re(z[j]),Im(z[j])],r,color=black,thickness=3:
> end do:
> pp0:=implicitplot(abs(x+I*y-zc)=R,x=-5..5,y=-5..5, color = black):
> display(seq(pp||i,i=0..2*n));
```



Рис. 1. Контур інтегрування та особливі точки функції

Знаходимо суму залишків в тих особливих точках, які потрапили всередину контуру інтегрування та отримуемо відповідь.

```
> zinC:=0: S:=0:
> for j from 1 to n do
>    if evalf(evalc(abs(z[j]-zc)))<R then
>      zinC:=zinC+1: S:=S+residue(f(z),z=z[j]):
>    end if: end do:
```

```
> `Кількість особливих точок всередині контура інтегрування` =zinC;
> Int(f(z),z=C..``)=2*Pi*I*S;}
```

Кількість особливих точок всередині контура інтегрування = 2

$$\int_{|z+2i|=3} \frac{1}{z^3(z^2+4)^2}\,dz = \frac{-1}{32} I\pi$$

Наступний приклад можна виконати цілком аналогічно попередньому прикладу 1, хоча це нераціонально в даному випадку. Підінтегральна функція має 4 особливі точки всередині контуру інтегрування, а z=3 і z=∞ — назовні.

Приклад 2. *Обчислити інтеграл* $\oint_{|z|=2} \frac{dz}{(z-3)(z^4+2)}$.

```
> with(plots): with(plottools):
> f:=z->1/((z-3)*(z^4+2)): zc:==0: R:=2; 'f(z)'=f(z); C:=abs(z-zc)=R:
```

$$f(z) = \frac{1}{(z-3)(z^4+2)}$$

```
>`Теорема Коші про залишки:`;
>Int('f(z)',z='C'..``)=2*Pi*I*Sum(res[z=z[k]]('f(z)'),k)
>Int(f(z),z=C..``)=2*Pi*I*Sum(res[z=z[k]](f(z)),k);
```

Теорема Коші про залишки:

$$\int_C f(z)dz = 2I\pi\left(\sum_k res_{z=z_k} f(z)\right)$$

$$\int_{|z|=2} \frac{1}{(z-3)(z^4+2)}\,dz = 2I\pi\left(\sum_k res_{z=z_k}\left(\frac{1}{(z-3)(z^4+2)}\right)\right)$$

```
> zzz:=[singular(f(z))]: n:=nops(zzz); `Особливі точки функції f(z):`;
> for j from 1 to n do  z[j]:=op(zzz[j][1])[2]; end do;
```

Особливі точки функції f(z):

$$z_1 := 3$$

$$z_2 := \frac{2^{(3/4)}}{2} + \frac{1}{2} I\, 2^{(3/4)}$$

$$z_3 := \frac{1}{2} I\, 2^{(3/4)} - \frac{2^{(3/4)}}{2}$$

$$z_4 := -\frac{2^{(3/4)}}{2} - \frac{1}{2} I\, 2^{(3/4)}$$

$$z_5 := -\frac{1}{2} I\, 2^{(3/4)} + \frac{2^{(3/4)}}{2}$$

```
> pp0:=implicitplot(abs(x+I*y-zc)=R,x=-4..4,y=-4..4, color= black,
   thickness =2):
> for j from 1 to n do
> pp||j:=textplot([Re(z[j]),Im(z[j])+0.2,"z"||j], align= ABOVE):
> pp||(n+j):=circle([Re(z[j]),Im(z[j])],0.03, color= black, thickness=3:
> end do:
> p1:=pointplot([Re(zc),Im(zc)], color=black, symbol= CROSS):
> p2:=textplot([Re(zc)+0.2,Im(zc)+0.1,"zc"],align= ABOVE):
> display(seq(pp||i,i=0..2*n),p1,p2);
```



Рис. 2. Контур інтегрування і особливі точки функції

```
> zinC:=0: S:=0: S1:=0: for j from 1 to n do
>   if evalf[5](evalc(abs(z[j]-zc))<R then
>      zinC:=zinC+1: S:=S+residue(f(z),z=z[j]): else
>      S1:=S1+residue(f(z),z=z[j]): end if; end do:
> `Кількість особливих точок всередині контуру інтегрування`=zinC;
> Int(‘f(z)’,z=‘C’..``)=2*Pi*I*S;
```

Кількість особливих точок всередині контуру інтегрування = 4

$$\int_C \frac{1}{(z-3)(z^4+2)}\,dz = (-0. - 0.02409638552\quad I)\,\pi$$

```
> `Кількість особливих точок зовні контуру інтегрування`=n-zinC;
> Int(‘f(z)’,z=‘C’..``)=2*Pi*I*(S1+residue(f(z),z= infinity));
```

Кількість особливих точок зовні контуру інтегрування = 1

$$\int_C \frac{1}{(z-3)(z^4+2)}\,dz = (-0. - 0.02409638552\quad I)\,\pi$$

```
> `Перевірка:`;
> Sum(res[z=z[k]]('f(z)'),k)+'res[z=infinity](f(z))'=
  =simplify(evalc(S+S1+residue(f(z),z=infinity)));
```

Перевірка:

$$\left(\sum_k \operatorname{res}_{z=z_k}\left(f(z)\right)\right)+\operatorname{res}_{z=\infty}\left(f(z)\right)=0$$

Звичайно треба шукати інтеграл, використовуючи тільки залишки в z=3 і z=∞.

```
> r1:=residue(f(z), z=z[1]): res[z=z[1]]('f(z)')=r1;
  r2:=residue(f(z), z=infinity): res[z=infinity]('f(z)')=r2;
```

$$\mathrm{res}_{z=z_1}\left(f(z)\right)=-\frac{1}{83}$$

$$\mathrm{res}_{z=\infty}\left(f(z)\right)=0$$

```
> Int(f(z),z=C..``)=-2*Pi*I*evalf(r1 + r2);
```

$$\int_C \frac{1}{(z-3)(z^4+2)}dz=-0.02409638554\,I\pi$$

Як можна побачити, в обох результатах співпадають 10 знаків після коми.

Виконання наведеної теореми в прикладі 1 нічого не спростило би, оскільки там 2 особливі точки всередині контуру інтегрування і 2 поза ним, включаючи z=∞.

Приклад 3. *Обчислити інтеграл* $\oint\limits_{|z|=2}\frac{dz}{z^{10}-2}$.

```
> with(plots): with(plottools):
> f:=z->1/(z^10-2): zc:=0: R:=2: zc:=0: R:=2:
> 'f(z)' = f(z); C:=abs(z-zc)=R:
```

$$f(z)=\frac{1}{z^{10}-2}$$

```
>`Теорема Коші про залишки:`;
>Int('f(z)',z='C'..``)=2*Pi*I*Sum(res[z=z[k]]('f(z)'),k);
>Int(f(z),z=C..``)=2*Pi*I*Sum(res[z=z[k]](f(z)),k);
```

Теорема Коші про залишки:

$$\int_C f(z)dz=2I\pi\left(\sum_k res_{z=z_k} f(z)\right)$$

$$\int_{|z|=2}\frac{1}{(z-3)(z^4+2)}dz = 2I\pi(\sum_k res_{z=z_k}(\frac{1}{(z-3)(z^4+2)}))$$

Знайдемо особливі точки даної функції й їх кількість.

```
> zzz:=[evalf(singular(f(z)))]: n:=nops(zzz):
> for j from 1 to n do  z[j]:=op(zzz[j][1])[2]: end do:
```

Малюємо контур інтегрування і особливі точки.

```
>pp0:=implicitplot(abs(x+I*y-zc)=R,x=-5..5,y=-5..5,color=black,thickness =2)
> for j from 1 to n do
> pp||j:=textplot([ Re(z[j]), Im(z[j]) + 0.2, "z" || j ],align =ABOVE ):
> pp||(n+j):=circle([Re(z[j]),Im(z[j])],0.03,color=black, thickness=3):
> end do:
> p1:=pointplot([Re(zc),Im(zc)],color=black,symbol=CROSS):
```

```
>p2:=textplot([Re(zc)+0.2,Im(zc)+0.1,"zc"\:],align=ABOVE)
> display(seq(pp||i,i=0..2*n),p1,p2);
```



Рис. 3. Контур інтегрування і особливі точки функції

Маємо десять особливих точок і всі вони потрапили всередину контуру інтегрування. Тому можемо знайти інтеграл, обчисливши лишок тільки в z=∞.

```
> Int(f(z),z=C..``)=-2*Pi*I*residue(f(z), z= infinity);
```

$$\oint_{|z|=2} \frac{1}{z^{10}-2}\,dz = 0$$

Приклад 4. *Обчислити інтеграл* $\oint_{|z|=1} z^2 \cdot e^{\frac{1}{z}}\,dz$.

Якщо спробувати скористатися попереднім алгоритмом і в цьому прикладі, то виникнуть певні труднощі. У даної функції особливими точками є нуль та нескінченно віддалена точка. Всередину контуру інтегрування потрапив нуль, в ньому – істотна особливість (див., наприклад, [7, приклади 7.3 і 7.5]). Тому лишок будемо шукати як c_{-1} коефіцієнт в розкладанні в ряд Лорана функції в колі нуля. Розклад в ряд Лорана отримують за допомогою команди `laurent`.

```
> with(numapprox): f:= z -> z^2*exp(1/z): `f(z)` = f(z);
```

$$f(z) = z^2\, e^{\left(\frac{1}{z}\right)}$$

Згадаємо формулу, що необхідна для підрахунку цього інтеграла.

```
> Int(f(z),z=C..``) = 2*Pi*I*c[-1];
```

$$\int_{|z|=1} z^2 e^{\frac{1}{z}} dz = 2I\pi c_{-1}$$

Здійснимо заміну змінної, підставивши $z = \frac{1}{t}$ в дану функцію. Розкладаємо в ряд Лорана отриману функцію і зробимо зворотну заміну змінної.

```
> `f(t)`=subs(z = 1/t,f(z)); `f(t)`=expand(laurent(rhs(%),t=0,6));
> `f(z)` = subs(t = 1/z, rhs(%));
```

$$f(t)=\frac{\mathbf{e}^t}{t^2}$$

$$f(t)=t^{-2}+t^{-1}+\frac{1}{2}+\frac{1}{6}t+\frac{1}{24}t^2+\frac{1}{120}t^3+O(t^4)$$

$$f(z)=z^2+z+\frac{1}{2}+\frac{1}{6z}+\frac{1}{24z^2}+\frac{1}{120z^3}+O\left(\frac{1}{z^4}\right)$$

Знайдемо коефіцієнт c_{-1}, який дорівнює залишку функції в істотно особливій точці 0.

```
>c[-1]:=coeff(convert(rhs(%),polynom),z^(-1)); Int(f(z),z)=2*Pi*I*C[-1];
```

$$c_{-1}:=\frac{1}{6}$$

$$\int_C z^2\,\mathbf{e}^{\left(\frac{1}{z}\right)}dz=\frac{1}{3}I\pi$$

Продемонструємо рішення задачі, аналогічної третій індивідуальній.

Приклад 5. *Обчислити інтеграл* $\oint_{|z|=2}\frac{z^6}{(z-3)(z^4+2)}dz$.

```
> with(plots): with(plottools):
> f:=z->z^6/((z-3)*(z^4+2)): zc:=0: R:=2: ‘f(z)’=f(z); C:=abs(z-zc)=R:
```

$$f(z)=\frac{z^6}{(z-3)(z^4+2)}$$

```
> `Теорема Коші про залишки.`;
> Int(‘f(z)’,z=‘C’..``)=2*Pi*I*Sum(res[z=z[k]](‘f(z)’),k)
> Int(f(z),z=C..``)=2*Pi*I*Sum(res[z=z[k]](f(z)),k);
```

Теорема Коші про залишки.

$$\int_C f(z)dz=2I\pi\left(\sum_k res_{z=z_k}f(z)\right)$$

$$\int_{|z|=2}\frac{z^6}{(z-3)(z^4+2)}dz=2I\pi\left(\sum_k res_{z=z_k}\left(\frac{z^6}{(z-3)(z^4+2)}\right)\right)$$

```
> `Особливі точки функції f(z):`; zzz:=[(singular(f(z)))]: n:=nops(zzz):
```

Особливі точки функції f(z):

$$zzz:=\left\{\left\{z=\frac{2^{\left(\frac{3}{4}\right)}}{2}+\frac{1}{2}I\,2^{\left(\frac{3}{4}\right)}\right\},\left\{z=-\frac{2^{\left(\frac{3}{4}\right)}}{2}+\frac{1}{2}I\,2^{\left(\frac{3}{4}\right)}\right\},\left\{z=-\frac{2^{\left(\frac{3}{4}\right)}}{2}-\frac{1}{2}I\,2^{\left(\frac{3}{4}\right)}\right\},\left\{z=\frac{2^{\left(\frac{3}{4}\right)}}{2}-\frac{1}{2}I\,2^{\left(\frac{3}{4}\right)}\right\},z=\infty,z=-\infty,z=3\right\}$$

```
> for j from 1 to n do
>    z[j]:=op(zzz[j][1])[2]: end do:
> pp0:=implicitplot(abs(x+I*y-zc)=R, x=-5..5,y=-5..5, color=black,
thickness=2): for j from 1 to n do
```

```
> if (z[j]<>infinity) and  (z[j]<>-infinity) then
> pp||j:=textplot([(Re(z[j]))+0.4,Im(z[j])-0.1,"z"||j], align= BELOW):
> pp||(n+j):=pointplot([Re(z[j]),Im(z[j])],color=black, symbol=CIRCLE,
symbolsize=14): else pp||j:=textplot([0,0,``]): pp||(n+j):=pp0:
> end if; end do;
> p1:=pointplot([Re(zc),Im(zc)],color=black,symbol= CROSS):
> p2:=textplot([Re(zc)+0.2,Im(zc)+0.1,"zc"],align=ABOVE):
> display(seq(pp||i,i=0..2*n),p1,p2);
```



Рис. 4. Контур інтегрування і особливі точки функції

```
> zinC:=0: S:=0: S1:=0: for j from 1 to n do
>   if evalf[5](evalc(abs(z[j]-zc)))<R then
>      zinC:=zinC+1: S:=S+residue(f(zz),zz=z[j]):  else
>      if z[j]<>infinity and z[j]<>-infinity then
>          S1:=S1+residue(f(zz),zz=z[j]): end if: end if; end do;
> `Кількість точок, які знаходяться в області`=zinC;
>  Int('f(z)',z=C..``)=2*Pi*I*evalf[5](S); d:=2*Pi*I*S:
> `Кількість точок, які не знаходяться в області`=n-zinC;
> Int('f(z)',z=C..``)=(-2*Pi*I*(S1+residue(f(zz),zz= infinity)));
```

$$\textit{Кількість точок, які знаходяться в області} = 4$$

$$\int_{|z|=2} f(z)dz = (-0. + 0.43372I)\pi$$

$$\textit{Кількість точок, які не знаходяться в області} = 3$$

$$f(z) = \frac{36}{83} I\pi$$

```
> `Перевірка:`; Sum(res[z=z[k]]('f(z)'),k)+res[z=infinity]('f(z)')=
simplify(evalc(S+S1 + residue(f(z),z=infinity)));
```

$$\textit{Перевірка:}$$

$$\left(\sum_k res_{z=z_k}(f(z)) \right) + res_{z=\infty}(f(z)) = 0$$

Висновки

В статті показано, як використовуючи пакет Maple в темі «Обчислення інтегралів від функції комплексної змінної за допомогою залишків», що вивчається студентами фізичних спеціальностей, зазвичай, у курсі «Теорія функцій комплексної змінної» («Комплексний аналіз») [5-7] або «Математична фізика», можна складні задачі з обчислення інтегралів звести до обчислення залишків в пакеті Maple, підвищити наочність задач. Отже, основні переваги використання математичних пакетів при вивчені курсу теорії функцій комплексної змінної можна сформулювати так: підвищення інтересу до предмета, організація індивідуальної навчальної діяльності студентів, скорочення непродуктивних витрат часу на допоміжні роботи, розвиток творчої активності студентів, підвищення унаочнення, виразності, доступності навчального матеріалу. Впровадження СКМ у процес навчання повинно бути метою для технічної освіти. Для підвищення ефективності даного процесу треба: орієнтуватися на використання єдиного програмного засобу в межах освітнього закладу; будувати курси, що базуються на математиці, з урахуванням використання відповідної системи; мати достатню кількість комп’ютерних лабораторій, що дозволять ефективно використовувати вибраний програмний засіб.

Список використаних джерел

1. Дьяконов В.П. Maple 9 в математике, физике и образовании. – М.:СОЛОН-Пресс, 2004. – 688 с.
2. Матросов А.В. Maple 6 Решение задач высшей математики и механики. – БХВ-Санкт-Петербург, 2001. – 528 с.
3. Сдвижков О.А. Математика на компьютере: Maple 8. – М.: СОЛОН-Пресс, 2003. – 176 с.
4. Garvan F. The Maple Book. – Chapman & Hall/Crc, 2002. – 463 p.
5. Mathews J.H., Howell R.W. Complex Analysis for Mathematics and Engineering – Jones and Bartlett Publishers , 2006. – 484 p.
6. Чудесенко В.Ф. Сборник заданий по специальным курсам высшей математики (типовые расчеты): Учеб. пособие для втузов. – М.: Высш. школа, 1983.– 112 с.
7. Парфёнова Н.Д. Комплексный анализ. – ХНУ, 2009. – 86 с.

ВЫЧИСЛЕНИЕ ИНТЕГРАЛОВ ОТ ФУНКЦИИ КОМПЛЕКСНОЙ ПЕРЕМЕННОЙ С ПОМОЩЬЮ ВЫЧЕТОВ В СИСТЕМАХ КОМПЬЮТЕРНОЙ МАТЕМАТИКИ

Парфёнова Н.Д., Клочко Т.В.

Аннотация

В статье приведена методика преподавания одной из важнейших тем курса комплексного анализа с использованием системы Maple; демонстрируется, как использование систем компьютерной математики облегчает оценивание работ студентов преподавателем при организации их самостоятельной работы.

Ключевые слова: Maple, математическое моделирование, информационные технологии в обучении.

EVALUATION OF INTEGRALS OF FUNCTIONS OF COMPLEX VARIABLE BY USING THE RESIDUES IN COMPUTER MATHEMATICS SYSTEMS

Parfyonova N.D., Klochko T.V.

Resume

The article describes the methodology of teaching one of the major topics of the course of complex analysis using the system Maple. It also demonstrates how the use of computer mathematics systems facilitates the evaluation of students by the teacher during organization of independent work.

Keywords: Maple, mathematical modeling, information technology in education.